
======TOGGLE_FF_TESTS==============
BEHAVIORAL MODELING OF FLIP FLIPS WITH FEEDBACK 
MAY BE DIFFICULT TO CONVERGE WITHOUT USING 
BREAK BEFORE MAKE LOGIC.   

TOGGLE_FF_TESTS
*=========Create_Signal==================
*.OPTIONS  GMIN=1p      METHOD=trap   ABSTOL=1m     TEMP=27   srcsteps = 1  gminsteps = 1 
*.OPTIONS  RELTOL=.001  ABSTOL=1p     VNTOL=1u      ITL4=500  ITL1=400 
*V_PULSE# NODE_P NODE_N DC     VALUE  PULSE( VINIT  VPULSE TDELAY TRISE  TFALL  PWIDTH PERIOD )
VSQ       SQ     0      DC     0      PULSE( 0      1      1u     1u     1u     1m     2m )

XTOGLE    SQ     OUT    TOGLE

.control
set       pensize = 2
tran      1u   10m      0        1u
plot      sq   out-.1
.endc

*=========TOGGLE============================================
*        _____________________________________
*       |                                     |
*       |   /    VC1 |\  _NVC1 /    VC2  |\   |  ___
*       |__/  _______| \/ \___/  ________| \__|_|OUT|
*          S1   _|_  | /\_/  S2   _|_    | /    |___|
*               ___  |/           ___    |/
*             C1 |              C2 |
*               _|_               _|_
*               ///               ///                  |\
*  ___   |\             VBF ____               DPE   __|B\__\S2
* |IN |__|3\________________|   \ PE                |  | /  /
* |___|  | / |           VLP| PE \___/\  /\  /\_____|  |/
*        |/  |/\  /\  /\____|    /     \/  \/   |   |
*               \/  \/   |  |___/      RLP2    _|_  |  |\
*               RLP1    _|_                    ___  |__|M\__\S1
*                       ___                CLP2 |      | /  /
*                   CLP1 |                     _|_     |/
*                       _|_                    ///
*                       ///
.SUBCKT   TOGLE   IN     OUT      
 BBUF     VBF     0      V =    u( v(IN )-.5 )
 RLP      VBF     VLP    1k
 CLP      VLP     0      5n  
 BAND     VPE     0      V =    u( u(v(VBF )-.5)*u(.5 -v(VLP ) ) -.5)
 RLP2     VPE     VLP2   .3k
 CLP2     VLP2    0      2n 
 BBRK     BRK     0      V =     1-u(v(VLP2 )-.2)
 BMAK     MAK     0      V =     u(v(VLP2 )-.9)
 S1       OUT     VC1    MAK     0        SW 
 S2       NVC1    VC2    BRK     0        SW 
 C1       VC1     0      10n
 C2       VC2     0      10n
 R1       VC1     0      10Meg
 R2       VC2     0      10Meg
 BINV     NVC1    0      V =     1-u(v(VC1)-.2)
 BOUT     OUT     0      V =     u(v(VC2)-.2)



.ENDS     TOGLE

.MODEL    SW      SW(    VT=.5 VH=.1  RON=1 ROFF=100MEG)

.end


